
ЛАБОРАТОРНАЯ РАБОТА № 1 

Изучение интерфейса среды моделирования FlexPDE  

 

1. ЦЕЛЬ РАБОТЫ 

Изучить интерфейс среды моделирования методом конечных 

элементов FlexPDE, закрепить теоретические знания, полученные при 

изучении дисциплины. 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Запуск FlexPDE. Для запуска FlexPDE достаточно дважды нажать этот 

значок  или находясь в проводнике Windows, перейти к папке, где 

установлен FlexPDE, и затем дважды щелкать на файле FlexPDE.exe. В 

любом случае, после запуска FlexPDE на экране появится следующее окно: 
 

 
Пункты меню имеют следующие функции: 

New 

Используйте этот пункт меню, чтобы создать новый дескрипторный 

файл. FlexPDE инициализирует описатель с наиболее общими заголовками 

раздела. В большинстве случаев,  будет более удобно создать новый 

описатель,  редактируя существующий, который является близким по смыслу 

к новой проблеме. 

Open 

Используйте этот пункт меню, чтобы открыть существующий файл для 

выполнения задачи или ее модификации. После выбора этого пункта меню 

появится стандартный диалог Windows Open_File. Переместитесь к папке, 

содержащей программу, которую требуется открыть. Например, переход к 

стандартной папке примеров "Samples\Steady_State\Heatflow" отобразит 

следующий экран: 
 

 
Рис. 1.1. 

Import 

Используйте этот пункт меню, чтобы импортировать текст программы 

из формата, отличного от используемого в системе FlexPDE. Единственная 

опция, доступная в настоящее время - "DXF", который импортирует файл, 

созданный в AutoCad. 

 

View 

Используйте этот пункт меню, чтобы повторно запустить графический 

вывод задачи FlexPDE,  которая была выполнена и закончена ранее. После 

выбора этого пункта меню появится стандартный диалог Windows Open_File. 

Выберите папку, содержащую нужный ".PGX" файл. Двойной щелчок на 

этом файле, или одиночный щелчок и нажатие Open открывают файл (см. 

следующий раздел, рассматривающий Сохраненные Графические Файлы для 

подробной информации). Вы можете рассматривать больше чем одну 

сохраненную задачу, и Вы можете открывать файлы для рассмотрения, в то 

время как другие задачи уже открыты, но Вы не должны открывать ту же 

самую задачу для одновременного рассмотрения и выполнения, так как 

могут происходить конфликты доступов к файлу. 

 

Exit 

Нажмите эту кнопку, чтобы закончить работу FlexPDE. Все открытые 

файлы и представления будут закрыты. Если изменения были сделаны и не 

сохранены, Вы будете запрошены. 

 

Нажмите эту кнопку, чтобы увидеть файл справки 

 

Редактирование и выполнение файлов. Вы можете открыть 

дескрипторный файл одним из двух путей:  

1) после запуска FlexPDE выбрав пункт меню "Open"; 

2) дважды нажимая на .PDE файле в Проводнике Windows.  

В любом случае, дескрипторный файл будет открыт, и появится окно 

редактирования. Также появится окно "состояния"  на левой стороне экрана. 

Например, предположите, что мы следуем за "Оpen" последовательно к 

папке " Samples\Steady_state\Heat_Flow " и выбираем "Heatbdry.pde". Окно 

редактирования выглядит следующим образом (рис. 1.2). Это - стандартное 

окно редактирования Notepad, которое показывает содержание выбранного  

 

Print 

Эта команда используется для вывода программы на печать. 

 

Run 

Начало выполнения открытого файл. Эта функция автоматически не 

сохраняет программу на диске. Вместо этого, программа будет сохранена во 

временном файле в текущей папке. Если произошел сбой, изменяемый файл 

может быть восстановлен из этого временного (см. описание сообщений в 

окне состояния во время выполнения задачи) файла. 

 

 
Рис. 1.2. 

 

После запуска программы (пункт меню Run Script или значок ) окно 

FlexPDE будет иметь вид, подобный приведенному на рис. 1.3. 

 

 
Рис. 1.3. 

 

Окно состояния (Status) 

В левой части находится окно "Status", которое содержит активные 

сообщения о состоянии решения и вид текущей вычислительной сетки. 

Формат напечатанных данных будет зависеть от вида решаемой задачи, но 

общие элементы таковы: 

Название(title) 

Число узлов сетки (nodes) 

Число конечных  элементов (cells) 

Число cтепеней свободы (Dof) 

Объем памяти, размещенный для задачи (Mem) 

Текущая оценка RMS  ошибки решения 

 

Другие пункты, которые могут появляться во время выполнения задачи: 

Текущее  время и временной шаг (timestep) 

Номер циклического повторения решения задачи (stage) 

RMS ошибка решения для последней итерации 

Номер итерации (Indecs) 

Состояние сходимости текущей итерации 

Сообщение текущего действия 

 

Меню наверху окна Status имеет две формы.  Во время выполнения 

задачи пункты меню: 

Stop 

Этот пункт остановит решение задачи как можно скорее, без попытки 

сохранить  результаты. 

 

Finish 

Этот пункт приведет к гладкому завершению решения в следующей 

удобном  точке остановки. Текущие решения будут закончены, но никакие 

построения новых сеток не будут начаты. Текущий шаг времени будет 

закончен, но никакие новые шаги по времени не будут начаты. Конечные 

(заключительные) графики будут записаны, и задача останется открытой, но 

неактивной. 

 

Когда задача неактивна, как в примере выше, пункты меню следуюшие: 

 

File 

Этот пункт вызывает раскрывающееся меню с несколькими функциями 

работы с файлами: 

Exit - закрывает дескрипторный файл, связанный с этой задачей, и 

стирает все ее окна на экране дисплея. Это  то же самое действие как кнопка 

'X' в верхнем правом угле окна состояния. 

Save - записывает программу на диск под активным названием(именем). 

Save as - вызывает стандартный Save_As диалог и записывает программу 

в новом файле на диске. 

Print all - посылает все видимые в настоящее время  окна на принтер. 

Modify - вызывает встроенный редактор, как будто программа только 

что была открыта. 

Continue - Этот пункт в настоящее время неактивен. Это станет частью 

приостанавливающегося / продолжающегося средства. 

Кнопка "Minimize" в окне Status может использоваться, чтобы 

минимизировать все графики, связанные с задачей. Кнопка "X"  в окне 

состояния может использоваться, чтобы закрыть проблему. 

 

Графические окна (Thumbnail). На правой стороне экрана - отдельные 

окна "Thumbnail" для каждого из ГРАФИКОВ (Plot) или МОНИТОРОВ 

(Monitor), требуемых описателем.  В установившихся проблемах  

МОНИТОРЫ будут отображаться в ходе выполнения задачи. Они будут 

заменены ГРАФИКАМИ, когда решение закончено. В неустановившихся 

задачах, все МОНИТОРЫ и ГРАФИКИ будут отображены одновременно и 

модифицированы, как это диктует спецификации описателя sequencing 

(последовательность). ГРАФИКИ будут записаны  на диске с расширением 

".PGX" для более позднего просмотра, МОНИТОРЫ записаны не будут. 

В задачах о собственных значениях будет иметься набор МОНИТОРОВ 

или ГРАФИКОВ для каждого требуемого режима. В других отношениях 

задачи о собственных значениях ведут себя как установившиеся проблемы. 

Каждый график  может быть изменен  перемещением границы  или 

максимизирован стандартной  кнопкой.  

 

Пункты меню: 

Setup - вызывает стандартный диалог "Установка принтера". 

Print - посылает окно принтеру, используя стандартный диалог 

"Печать". 

Black_Print - посылает окно принтеру со всеми цветами, 

отображенными к черному. Это полезно для создания viewgraphs, или в тех 

случаях, когда стандартный Windows дает несоответствующее greyscale 

отображение  

 

Export - вызывает подменю, которое позволяет выбрать формат для 

экспорта графика к другим приложениям. В настоящее время  единственная 

опция - "EMF"( Windows Enhanced Metafile). Этот формат позволяет графику 

быть импортированным в любое приложение, которое поддерживает формат 

Windows Enhanced Metafile ( типа Word MicroSoft ). 

 

Графическое маркирование (Plot Labeling). Типичный КОНТУРНЫЙ 

график имеет следующий вид: 

Рис. 1.4. 

 

Наверху дисплея в левом углу название задачи, в правом углу 

время/дата  выполнения, а также версия FlexPDE, которая выполнила 

вычисления. Внизу страницы - резюме прикладной статистики, подобной 

показанному в окне состояния: 

Название 

Число циклов модернизации сетки (gridding), выполненных на данный 

момент 

Полиномиальный порядок базисных функций (p2 = квадратный, p3 =     

кубический) 

Число узлов вычисления 

Число ячеек вычисления 

Оцененная ЭФФЕКТИВНАЯ ВЕЛИЧИНА относительной погрешности 

решения 

В циклических задачах номер цикла будет сообщен (stage). 

В задачах о собственных значениях номер моды будет сообщен. 

В неустановившихся задачах текущее  время и временной шаг (timestep) 

будут сообщены. 

По умолчанию  FlexPDE вычисляет интеграл по отображенной кривой, и 

это значение сообщается как "Интеграл". 

Любые требуемые сообщения будут появляться в нижней строке. 

 

Типичный график функции имеет следующим вид: 

Рис. 1.5. 

 

Здесь также выводятся все данные  графика и  эскиз области. Для 

граничных графиков, соединения границы перечисляются на эскизе. 

 

Просмотр сохраненных графических файлов. Всякий раз, когда 

задача выполнена FlexPDE, графический вывод, сделанный разделом PLOT 

описателя, записывается в файле с расширением ".PGX". Эти файлы могут 

позже рассматриваться FlexPDE без  повторного запуска программы. Пункт  

"View" на основном меню FlexPDE выбирает эту функцию. Стандартный 

Windows-диалог "Оpen" используется для выбора одно из доступных файлов. 

Как только файл выбран, первый блок графиков будет отображен, подобно 

этому: 

 

 
Рис. 1.6. 

 

Слева - окно "Status", очень подобное тому, который появляется в ходе 

расчета. Не вся информация во время выполнения будет появляться здесь, а 

только те пункты, которые нужны для идентификации данной группы 

графиков. В стационарных задачах  все окна PLOT будут отображены. Если 

задача циклическая, то каждый цикл будет появляться в отдельной группе. В 

неустановившихся  задачах  каждое графическое время, указанное в разделе 

PLOT описателя, формирует группу. Окно Status теперь имеет другое меню, 

чем  в течение выполнения задания. Новые пункты такие: 

 

Restart - этот пункт заставит FlexPDE возвращаться началу файла и 

отображать первую группу. 

Previous - этот пункт заставит FlexPDE копировать и восстанавливать 

изображение предыдущей группы. Если не имеется никаких более ранних 

групп,  сообщение будет появляться. 

Next - используйте этот пункт, чтобы продвинуться к следующей группе 

графиков в файле. Если не имеется больше групп,  сообщение будет 

появляться. 

Print_All - этот  пункт  все видимые окна  посылает принтеру. 

Индивидуальные окна могут быть напечатаны из их собственных строк 

меню. 

Используйте "X" кнопку, чтобы закрыть файл. 

 
3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного моделирования методом конечных элементов FlexPDE, объектом 

исследования является интерфейс этой программной среды – команды меню 

и панели инструментов.  
 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Открыть файлы с примерами, указанными в теоретической части, 

запустить программу и ознакомиться с окном результатов моделирования. 
3. Подготовить отчет о выполнении работы. 
 
 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который описания 

основных команд меню и инструментальных панелей системы FlexPDE. 
 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой метод численногно решения используется в системе FlexPDE? 
2. Опишите пункты меню верхнего уровня системы FlexPDE. 
3. Какая информация отображается в окне состояния? 
4. Какие команды используются для просмотра сохраненных 

графических файлов. 
5. Какой вид имеет контурный график в системе FlexPDE? 

 

7. ИСПОЛЬЗУЕМАЯ ЛИТЕРАТУРА 
1. Дворецкий С.И., Ермаков А.А., Иванов О.О., Акулинин Е.И. 

Компьютерное моделирование процессов и аппаратов пищевой, био- и 

химической технологии в среде FlexPDE: Учебное пособие. - Тамбов: 

Издательство ТГТУ, 2006. - 72 с.    http://window.edu.ru/resource/586/38586 

2. Маслов Ю.А., Меринов И.Г., Рябов Н.О. Моделирование 

теплогидравлических процессов в реакторных установках и элементах 

теплообменного оборудования ЯЭУ: лабораторный практикум. – М.: МИФИ, 

2008. – 156 с.   http://www.studfiles.ru/preview/413352/page:8/ 

3. FlexPDE 6: User Guide-

http://www.pdesolutions.com/help/index.html?user_guide.html 

4. FlexPDE - численное решение систем дифференциальных уравнений 

в частных производных - http://www.1024.ru/studentam/flexpde 

http://window.edu.ru/resource/586/38586
http://www.studfiles.ru/preview/413352/page:8/
http://www.pdesolutions.com/help/index.html?user_guide.html
http://www.1024.ru/studentam/flexpde


13 

ЛАБОРАТОРНАЯ РАБОТА № 2 

Основы программирования в среде моделирования FlexPDE  

 

1. ЦЕЛЬ РАБОТЫ 

Изучить типы разделов программы, основные команды, используемые 

для описания решаемой задачи, начальных и граничных условий, геометрии 

расчетной области, а также для вывода полученных результатов, закрепить 

теоретические знания, полученные при изучении дисциплины. 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Создание дескрипторных файлов. Перед представлением задачи для 

анализа FlexPDE Вы должны подготовить  дескрипторный файл (программу). 

Для записи программ используется простой для изучения естественный язык, 

первоначально разработанный Робертом  Нельсоном. Небольшое  число 

расширений было добавлено, чтобы увеличить возможности FlexPDE. 

По сравнению с  языками программирования  естественный язык, 

используемый в прикладных программах, намного более простой. 

Большинство  студентов, инженеров и ученых, которые имели по крайней 

мере вводный курс  дифференциальных уравнений в частных производных, 

могут достаточно быстро овладеть языком, чтобы готовить простые 

прикладные дескрипторные файлы и начинать решать свои задачи. 

Набор команд FlexPDE может рассматриваться как язык стенографий 

для создания конечно-элементных моделей. Команды обеспечивают 

информацию, необходимую  FlexPDE, чтобы транслировать числовые 

процессы для решения задачи. 

Язык FlexPDE  описательный, а не процедурный. Пользователь 

описывает, как различные компоненты системы связаны друг с другом. Он не 

задает последовательность шагов, которые нужно выполнить для 

формирования решения. На  основании связей между переменными задачи, 

FlexPDE сам составляет последовательность шагов, необходимых для 

решения. FlexPDE делает различные предположения относительно свойств 

переменных задачи, в зависимости от раздела объявления.   

Например, если переменная названа в разделе VARIABLES, то принято:  

1) эта переменная - скалярное поле, которое принимает значения в 

области проблемы,  

2) эта переменная будет смоделирована конечной интерполяцией  между 

узлами сетки,  

3) значения переменной непрерывны в области,  

4) дифференциальное уравнение в частных производных, описывающее 

поведение переменной, будет удовлетворено. 

Если переменная появляется в разделе DEFINITIONS, то относительно 

названной величины принимается: 

1) эта величина вспомогательная для PDE системы,  

2) эта величина может быть разрывна  в области, 

3) эта величина  обязательно удовлетворяет условиям любого PDE. 

 

Прикладные дескрипторные файлы для использования с FlexPDE 

наиболее легко подготовить и отредактировать, используя встроенный 

редактор FlexPDE . Для открытия нового дескрипторного файла просто 

нажимают NEW в  строке меню FlexPDE. Чтобы редактировать 

существующий описатель, вместо этого нажмите OPEN. Удобный способ 

создавать новый описатель состоит в том, чтобы начать с существующего 

описателя для подобной проблемы и изменять его, чтобы удовлетворить 

новым прикладным условиям. 

Встроенный редактор FlexPDE идентичен редактору Notepad и 

производит текстовый файл ASCII без любых вставленных символов. 

Дескрипторные файлы могут также быть подготовлены, используя любой 

редактор текста ASCII или любой редактор, способный к экспорту 

текстового файла ASCII. 

 

Имена  и расширения файла. Прикладной дескрипторный файл может 

иметь любое название, которое является совместимым с Windows. Лучше 

выбрать имя, которое является наглядным для задачи. Прикладные 

дескрипторные файлы должны иметь расширение '.PDE '. При сохранении 

файла с использованием встроенного редактора,  FlexPDE автоматически 

прибавит расширение '.PDE '. При использовании отдельного или внешнего 

редактора, убедитесь, что дали файлу '.PDE ' расширение вместо заданного 

по умолчанию данным редактором. 

 

Структура  дескрипторных файлов. Прикладные описатели 

организовывают задачу, разделяя их на разделы, связанные между собой. 

Каждый раздел озаглавливается  собственным именем, включает одну или 

большее количество инструкций, которые определяют проблему. 

Разрешенные названия разделов: 

Title    - определяет  заголовок 

Select    - устанавливает различные опции и средства управления 

Coordinates   - определяет систему координат 

Variables  - называет  переменные 

Definitions   - определяет вспомогательные величины и параметры 

Initial values   - устанавливает начальные значения переменных 

Equations  - определяет систему дифференциального уравнения в 

частных производных 

Constraints   - определяет необязательные интегральные связи 

Resolve   - выбирает дополнительный контроль(управление) сеткой 

Extrusion   - расширяет область на три измерения 

Boundaries   - описывает 2- или проектируемый 3-мерный 

домен(область) и задает граничные условия 

Time    - определяет временной интервал  

Monitors   - выбирает  промежуточный графический дисплей  

Plots    - выбирает конечный(заключительный) графический 

дисплей 

Histories   - выбирает итоговые по времени дисплеи 

End    - идентифицирует конец описателя 

 

Число разделов, используемых в программе, может изменяться, 

выполняя только требование, чтобы все файлы содержали разделы 

переменных, уравнений, границ и конечный раздел. 

В то время как существует некоторая гибкость в размещении этих 

разделов, предлагается, что пользователь твердо придерживается 

упорядочения, описанного выше. Definitions и Select могут появляться 

больше чем однажды.  Поскольку описатели динамически обрабатываются 

сверху вниз, они не могут содержать ссылки вперед. Определения могут 

содержать переменные и другие определенные имена, если эти переменные и 

имена были определены в предшествующем разделе или предварительно в 

том же самом разделе. 

 

Прикладной дескрипторный формат. FlexPDE  предполагает 

использование следующей структуры для всех прикладных описателей 

(впрочем не строго требуемой):  

                Раздел 1 

   Инструкция 

  Раздел 2 

   Инструкция 1 

   Инструкция 2 

    * 

    * 

  Раздел 3 

   Инструкция 1 

   Инструкция 2 

    * 

    * 

Этот формат прост и для человека, готовящего файл, и для других, 

чтобы читать и понимать. 

 

Чувствительность к  выбору регистра. За исключением литеральных 

символьных строк, которые воспроизводятся точно так , как  они появляются 

в прикладном описателе, слова, символы и другие текстовые элементы, 

используемые в прикладных описателях,  нечувствительны к регистру. 

Символы верхнего регистра и символы строчных букв эквивалентны. 

Включаемые файлы. FlexPDE поддерживает механизм языка C 

включения внешних файлов в прикладной описатель. Инструкция  

 

  #INCLUDE " <Имя файла> "   

 

заставит названный файл быть включенным целиком в описатель на месте 

этой инструкции 

Если файл не находится в той же самой папке, что и  описатель, то 

нужно указать полный путь к файлу. Включающаяся инструкция может быть 

помещена куда-нибудь в описателе, но для удобочитаемости, она должна 

быть помещена в отдельную строку. 

Это средство может использоваться, чтобы создать дескрипторные 

файлы большие, чем предел 32КБ, наложенный Windows, или вставлять 

общие группы определений в несколько описателей. 

 

 

Простой пример. Как пример дескрипторного файла FlexPDE, мы 

создадим модель  Нeatflow (тепловой поток) для квадратной области. 

Уравнение теплового потока   

      div (K*grad (T)) + Source = 0 

Это уравнение удовлетворяется функцией  

      T = const - x ** 2 - y ** 2, 

если K постоянно и Source = 4*K. 

Мы определяем квадратную область(регион) материала удельной 

проводимости K = 1, с однородным источником  Source=4 единиц теплоты на 

единицу площади. Мы далее определяем граничное значение 

      T = 1 - x ** 2 - y ** 2 

Так как  аналитическое решение известно, мы можем сравнивать 

точность FlexPDE решения. 

Текст описания задачи следующий: 

{****************************************************************    

SIMPLE.PDE  

Этот пример демонстрирует самое простое приложение FlexPDE  

к проблеме стационарного теплопереноса (Heatflow).   

     

****************************************************************} 

Title "Simple Heatflow " 

 

Variables 

T (range=1)   { "T" - переменная системы, range - предполагаемый 

интервал изменения} 

 

Definitions 

K = 1   { удельная проводимость} 

Source = 4   { источник теплоты} 

Texact = 1-x ** 2-y ** 2   {Для сравнения определяется точное решение} 

 

Initial value 

T = 0   {начальные данные, необязательны в линейных установившихся 

проблемах, но необходимы для   нестационарных или нелинейных систем} 

 

Equations  {(heatflow) уравнение теплового потока} 

div (K*grad (T)) + Source = 0        

 

Boundaries       {Определение области и граничных условий} 

Region 1   {... только одна область (регион)} 

 Value (T) =Texact  {Определяет  условие Дирихле на границе  Т= 

точному решению} 

  Start (-1, -1)    {Определяет отправную точку} 

  line to (1, -1)    {Идет граница} 

  to (1,1)  

  to (-1,1)  

Finish    {Возвращает границу к отправной точке} 

         

Monitors  

 Contour (T)   {Показывает линии постоянной температуры в процессе 

решении} 

 

Plots   { Показывает окончательные графики, записываются в файл при 

завершении} 

 

Contour (T)    { линии постоянной температуры } 

Surface (T)    {поверхностный график } 

Contour (T - Texact) as "Error"   {Отображает ошибку решения} 

Vector (-dx (T), -dy (T)) as "Heat Flow"   { векторный график 

теплового потока} 

 

End   {Конец дескрипторного файла} 

 

Описание разделов программы. 

Комментарии 

Текст, заключенный в фигурные скобки    {   } 

Текст после восклицательного знака ! 

 

COORDINATES { задание типа системы координат} 

CARTESIAN1 - 1D система координат (ось 'X‘) 

CYLINDER1   - 1D цилиндрическая система координат (ось 'R‘) 

SPHERE1        - 1D сферическая система координат (ось 'R‘) 

CARTESIAN2 - 2D система координат (оси 'X' и 'Y‘) 

XCYLINDER - 2D цилиндрическая система координат (осевая 

координата 'Z' направлена горизонтально, а радиальная 'R' – вертикально) 

YCYLINDER  2D - 2D цилиндрическая система координат (радиальная 

координата 'R' направлена горизонтально, а осевая 'Z‘ – вертикально) 

CARTESIAN3  - 3D система координат (оси 'X', 'Y' и 'Z‘) 

 

VARIABLES        { переменные } 

u, v, Pot, Temp    { имена используемых переменных } 

Все названия, появляющиеся в этом разделе будут представлены 

конечноэлементной аппроксимацией по прикладной сетке. Каждая 

переменная определяет непрерывное скалярное поле по области. Далее 

принимается, что каждая переменная будет сопровождаться 

дифференциальным уравнением в частных производных, перечисленным в 

разделе EQUATIONS. 

При назначении имен зависимым переменным, применяются следующие 

правила: 

1) имена переменных должны начаться с буквы. 

2) имена переменных должны быть уникальными и отличаться от 

символа t, который зарезервирован для переменной время. 

3) имена переменных могут иметь любую длину и любую комбинацию 

символов, цифр и-или символов, отличных от зарезервированных слов. 

4) имена переменных не должны содержать любые разделители. 

Составные имена могут быть сформированы с '_' символом (например 

temperature_celsius). 

5) имена переменных не могут содержать '- ', который зарезервирован 

для знака "минус". 

 

SELECT              { параметры численного решения } 

NGRID 20 - определяет число ячеек сетки в максимальном измерении. 

NODELIMIT 10000 - определяет максимальное число узлов 

NONLINEAR AUTOMATIC - выбирает нелинейный метод для решения 

NONSYSMMETRIC Automatic - выбирает несимметричный метод решения 

QUADRATIC On - используют квадратичные базисные функции 

REGRID On - по умолчанию FlexPDE осуществляет адаптивное 

усовершенствование сетки. Этот селектор может использоваться, чтобы 

выключить это и использовать фиксированную сетку. 

 

DEFINITIONS    { параметры решаемой задачи } 

  a 

  L=10 

  Q=exp(-a*t)*(x^2+y^2)^0.5 

Раздел DEFINTIONS используется, чтобы объявить и назначить 

названия специальных постоянных, коэффициентов и функций, 

используемых в прикладной задаче. При назначении названий к 

определениям применяются следующие правила: 

1) Названия определений должны начаться с буквы.  

2) Имена определений должны быть уникальными и отличаться от 

символа t, который зарезервирован для переменной время. 

3) Названия определений могут иметь любую длину и любую 

комбинацию символов, цифр и символов, отличных от зарезервированных 

слов, координатных имен или имен переменных. 

4) Названия определений не должны содержать любые разделители. 

Составные названия могут быть сформированы с '_' символ (например 

temperature_celsius). 

5) Названия определений не могут содержать '- ', который 

зарезервирован для знака "минус". 

Обычно, когда определение объявлено, то применением оператора 

назначения "=" задается его значение или выражение. Определения - 

динамические элементы и когда значение назначено, это будет только 

начальное значение и оно может быть модифицировано, в случае 

необходимости, в соответствии с решением. 

 

EQUATIONS { дифференциальные уравнения, для каждой переменной} 

Раздел EQUATIONS используется, чтобы перечислить 

дифференциальные уравнения в частных производных, которые определяют 

зависимые переменные задачи. Должно иметься одно уравнение для каждой 

зависимой переменной, перечисленной в разделе VARIABLES. Взаимно-

однозначные отношения установлены между переменными, уравнениями и 

граничными условиями. 

Если в задаче используется несколько переменных (уравнений),  то 

перед каждым уравнением требуется указать, какой переменной это 

уравнение соответствует. 

Одно уравнение 

div(grad(u))=0 

 

Два уравнения 

u:        div(grad(u))=0 

Temp: dxx(Temp)+dyy(Temp)=0 

 

Правила записи производных: 

dU/dx  => dx(U), dU/dy  => dy(U) , dU/dt  => dt(U) 

d2U/d2x  => dx(dx(U)) = dxx(U) 

В цилиндрических и в сферических координатах определены операторы 

- DR, DZ, DRR, DRZ, DZR и DZZ 

 

CONSTRAINTS { интегральные ограничения } 

Раздел CONSTRAINTS, который является необязательным, 

используется, чтобы применить интегральные связи к системе. Эти связи 

могут использоваться, чтобы устранить неоднозначность, которая иначе 

произошла бы в установившихся системах, или когда граничные условия 

содержат только производные зависимых переменных. Раздел 

CONSTRAINTS обычно содержит один или большее количество инструкций 

формы 

INTEGRAL(argument) = Expression 

CONSTRAINTS не должны использоваться с системами 

установившегося вида, которые однозначно определены их граничными 

условиями, или в системах, зависящих от времени. 

CONSTRAINTS не могут использоваться, чтобы предписать локальные 

требования (типа positivity) к переменной. 

 

BOUNDARIES      { описание границы расчетной области и граничных 

условий } 

  REGION 1           { область, заполненная одним материалом } 

    START(0,0)       { начальная точка границы области } 

    BOUNDARY CONDITION { граничные условия для 1-го участка 

границы} 

    LINE TO (1,0)   { следующий участок границы области } 

 BOUNDARY CONDITION { граничные условия для I-го участка 

границы} 

TO CLOSE           { замыкание контура, задающего границу области} 

 

Раздел BOUNDARIES используется, чтобы описать двумерную область 

или проекцию трехмерной области на основную плоскость, и задать 

граничные условия и внешние источники на границах задачи. Каждая 

прикладная программа должна иметь раздел BOUNDARIES. В пределе 

BOUNDARIES физическая область делится на части REGION, FEATURES и 

EXLUDE (подразделы). Каждая программа должна иметь по крайней мере 

один подраздел REGION. FEATURES и EXLUDE подразделы 

необязательные.  

REGION 1           { область, заполненная одним материалом } 

    START(0,0)       { начальная точка границы области } 

    BOUNDARY CONDITION { граничные условия для 1-го участка 

границы} 

    LINE TO (1,0)   { следующий участок границы области } 

 BOUNDARY CONDITION { граничные условия для I-го участка 

границы} 

TO CLOSE           { замыкание контура, задающего границу области} 

 

Подразделы REGIONS используются, чтобы описать замкнутые области, 

которые составляют физическую геометрию задачи в двумерной плоскости, 

или двумерные проекции на основную плоскость в трехмерных задачах. 

Подразделы REGIONS включают области и подобласти с определимыми 

материальными параметрами. Подразделы REGIONS сформированы таким 

образом, что начинаются с зарезервированного слова START, 

сопровождаемого физическими координатами начальной точки, затем 

следует идти вдоль границы области, описывая их прямолинейными 

отрезками или дуговыми сегментами, и замыкается область 

зарезервированным словом CLOSE. Координаты имеют стандартную 

математическую форму (X, Y).  

Прямолинейные отрезки задаются в следующем виде LINE TO (X, Y) 

Дуговые сегменты имеют одну из следующих форм: 

ARC to (X1, Y1) to (X2, Y2) 

ARC (RADIUS=R) to (X, Y) 

ARC (CENTER=X1, Y1) to (X2, Y2) 

ARC (CENTER=X1, Y1) ANGLE=Fi 

Fi - есть угол, измеренный в градусах, с учетом стандартного 

соглашения, по которому положительные углы отсчитываются против хода 

часовой стрелки, а отрицательные углы - по ходу часовой стрелки.  

Чтобы избегать разночтений с тригонометрическими функциями, 

которые берут углы в радианах, спецификаторы RADIANS и DEGREES 

могут следовать за углом. 

 Когда используется ARC (CENTER=X1, Y1) to (X2, Y2) и CENTER 

(X1, Y1) не равноотстоит от начального и конечного пунктов, генерируется 

сегмент эллиптический дуги.  

Когда используются последовательные сегменты LINE, 

зарезервированное слово LINE может не повторяться: 

LINE to (X1, Y1) to (X2, Y2) to (X3, Y3) to .. 

 

EXLUDES (исключения). Подразделы EXLUDES используются, чтобы 

описать замкнутые области, который имеют оверлейные (пересекающиеся) 

части в одном или большем количестве подразделов REGIONS. Область, 

описанная подразделом EXLUDES, исключается из системы. Подразделы 

EXLUDES, должен следовать за подразделами REGIONS, которым они 

перекрываются. Подразделы EXLUDES формируются тем же самым 

способом, как и подразделы REGIONS, и могут использовать LINE и ARC 

сегменты. 

 

FEATURE (элементы). Подразделы FEATURE используются, чтобы 

описать незамкнутые объекты, которые не включают подобласть с 

заданными материальными параметрами. Подразделы FEATURE 

формируются тем же самым способом, как и подразделы REGIONS, и могут 

использовать LINE и ARC сегменты.  

Подразделы FEATURE не заканчиваются резервным словом CLOSE. 

Подразделы FEATURE используются, когда задача имеет внутренние 

линейные источники; когда желательно вычислить интегралы по 

нерегулярному пути; или когда требуется явное управление вычислительной 

сеткой. 

Последовательность описания областей. Рекомендуется, чтобы все 

области REGIONS были описаны раньше, чем EXLUDES или FEATURE, и 

чтобы все EXLUDE подразделы быть описаны до задания FEATURE.  

Рекомендуется, чтобы первая область REGIONS была сформирована 

движением вдоль внешней границы расчетной области задачи. Для 

перекрывающихся областей, свойства в зоне перекрытия определяются 

свойствами последней области, в которой они были определены. 

 

Нумерация областей. Подразделы REGIONS, EXLUDES и FEATURE 

могут иметь номера и имена. Когда номера назначены, они должны быть в 

порядке возрастания, начиная с первого. Рекомендуется номера всегда 

назначать. Когда имена назначены, они должны иметь форму цитируемой 

строки и должны быть помещены немедленно после зарезервированного 

слова REGIONS, EXLUDES или FEATURE. Назначенные имена должны 

быть уникальными для REGIONS, EXLUDES, и FEATURE. Назначение 

имени области полезно, когда желательны графики или объемные интегралы 

ограниченные областью. 

Граничные условия (Boundary conditions). Cледующие формы 

граничных условий могут применяться: 

VALUE(VARIABLE)        =  Выражение 

NATURAL(VARIABLE)  =  Выражение 

LOAD(VARIABLE)          =  Выражение 

NEUMANN(VARIABLE) =  Выражение 

DNORMAL(VARIABLE) =  Выражение 

DTANGENTIAL(VARIABLE) =  Выражение 

NOBC(VARIABLE) 

 

VARIABLE – имя переменной, обозначенное в спецификации условия, 

идентифицирует уравнение, к которому это граничное условие должно 

примениться. 

VALUE – задает значение переменной (граничное условие первого 

рода).  NATURAL (LOAD – синоним) – задает значение потока (граничное 

условие второго рода) 

NEUMANN (DNORMAL – синоним) – задает значение производной по 

внешней нормали. (В частном случае уравнения DIV (Grad (U)) +f=0, 

граничные условия NEUMANN и NATURAL эквивалентны) .  

DTANGENTIAL – задает касательную производную 

NOBC -  отменяет предварительно заданное граничное условие на 

текущем пути (эквивалентно NATUAL(VARIABLE) =0) 

 

TIME { для нестационарных задач } 

Раздел TIME используется в нестационарных задачах, чтобы определить 

диапазон времени, в котором должна быть решена задача. Поддерживаются 

следующие альтернативные формы: 

FROM time1 TO time2 

FROM time1 INCREMENT TO time2 

FROM time1 TO time2 increment 

где: Time1 - начальное временя 

       Time2 – конечное время 

       Increment – шаг по времени 

Заданный по умолчанию начальный шаг времени - 1e-4 * (time2-time1). 

 

MONITORS         { отображение хода решения } 

PLOTS                  { отображение результатов решения } 

Раздел Monitors, который является необязательным, используется, 

чтобы перечислить графические дисплеи, желательные при промежуточных 

шагах, в то время как задача решается. 

Раздел Plots, который является необязательным, используется, чтобы 

перечислить графические дисплеи, желательные при завершении проблемы 

или стадии (в случае многократного повторения решения). Графики 

отличаются от мониторов также тем, что они записываются в файл с 

расширением .PGX для последующего рассмотрения после выполнения 

задачи. 

Инструкции Plots и Monitors имеют одинаковую форму и функции. 

СONTOUR (arg) – выводит двумерный контурный график линий 

постоянного уровня параметра arg 

GRID (arg1, arg2) – выводит сетку конечных элементов с узловыми 

координатами 

SURFACE(arg) – выводит квазитрехмерную поверхность, которая 

отображает параметр ARG по вертикали  

VECTOR(arg1, arg2) – выводит векторный график. 

 
3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, объектом исследования 

разделы и команды описания задач. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Выполнить текстовые расчеты по программам указанным 

преподавателем. 
3. Ввести в систему FlexPDE текст программы, приведенный в 

теоретической части, и выполнить расчеты. Внести изменения в текст 
программы по заданию преподавателя 

4. Подготовить отчет о выполнении работы. 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания основных разделов описания задач в среде FlexPDE, текст 
прикладной программы и полученные результаты моделирования. 

 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Что такое дескрипторный файл? 
2. Какое расширение имеют дексрипторные файлы системы FlexPDE? 
3. Какие разделы могут быть включены в дескрипторный файл? 
4. Что описывается в разделе Definitions? 
5. Какая команда используется для задания начальной точки границы 

двумерной расчетной области? 
 

7. ИСПОЛЬЗУЕМАЯ ЛИТЕРАТУРА 
1. Дворецкий С.И., Ермаков А.А., Иванов О.О., Акулинин Е.И. 

Компьютерное моделирование процессов и аппаратов пищевой, био- и 

химической технологии в среде FlexPDE: Учебное пособие. - Тамбов: 

Издательство ТГТУ, 2006. - 72 с.    http://window.edu.ru/resource/586/38586 

2. Маслов Ю.А., Меринов И.Г., Рябов Н.О. Моделирование 

теплогидравлических процессов в реакторных установках и элементах 

теплообменного оборудования ЯЭУ: лабораторный практикум. – М.: МИФИ, 

2008. – 156 с.   http://www.studfiles.ru/preview/413352/page:8/ 

3. FlexPDE 6: User Guide-

http://www.pdesolutions.com/help/index.html?user_guide.html 

4. FlexPDE - численное решение систем дифференциальных уравнений 

в частных производных - http://www.1024.ru/studentam/flexpde 

http://window.edu.ru/resource/586/38586
http://www.studfiles.ru/preview/413352/page:8/
http://www.pdesolutions.com/help/index.html?user_guide.html
http://www.1024.ru/studentam/flexpde

ЛАБОРАТОРНАЯ РАБОТА № 3 

Моделирование стационарного теплопереноса 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач стационарного теплопереноса в 

среде FlexPDE, исследовать влияние параметров системы на распределение 

температуры и величину теплового потока, закрепить теоретические знания, 

полученные при изучении дисциплины. 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Стационарный теплоперенос в двумерной области Ω, ограниченной 

границей Γ (рис. 3.1) описывается уравнением Лапласа: 

  0graddiv T  (3.1) 

где 
y

A

x

A yx









Adiv  - оператор дивергенции векторного поля А; 

yx 







 jigrad   - оператор градиента; T(x,y) – температура, значения 

которой зависят от координат (x, y). 

 

 
 

Рис. 3.1. 

 

Уравнение (3.1) можно записать в следующем виде: 

Будем считать, что участки границы, обозначенные нижним индексом И 

(ГИ1, ГИ2, ГИ3 и ГИ4) являются теплоизолированными, то есть тепловой поток 

через эти границы равен нулю: 

0

И4И3И2И1  , , ,






n

T
 (3.3) 

где n – единичный вектор внешней нормали к соответствующему участку 

границы области. 

Верхняя (ГН) и нижняя (ГС) границы являются изотермическими, то есть 

на них поддерживается постоянная температура: 

CH
CH

  , TTTT 
  (3.4) 

Для численного решения рассматриваемой задачи в среде FlexPDE 

может быть использована следующая программа 

TITLE Steady State 2D Heat Transfer 

COORDINATES 

  cartesian2 

VARIABLES 

Temp 

DEFINITIONS 

L=5        L1=2 

H=2       H1=4 

xP1=0    yP1=0 

xP2=L   yP2=0 

xP3=L   yP3=H 

xP4=L1 yP4=H 

xP5=L1 yP5=H1 

xP6=0 yP6=H1 

TempH=10  TempC=0 

FluxC=LINE_INTEGRAL(normal(grad(Temp)),'C') 

FluxH=-LINE_INTEGRAL(normal(grad(Temp)),'H') 

EQUATIONS 

Temp:  Div(grad(Temp)) = 0 

BOUNDARIES 

REGION 1 

    START(xP1,yP1) 

      VALUE(Temp)=TempC 

     LINE TO(xP2,yP2) 

      NATURAL(Temp)=0 

    LINE TO (xP3,yP3) 

      NATURAL(Temp)=0 

    LINE TO (xP4,yP4) 

      NATURAL(Temp)=0 

    LINE TO (xP5,yP5) 

      VALUE(Temp)=TempH 

    LINE TO (xP6,yP6) 

      NATURAL(Temp)=0 

    LINE TO CLOSE  

Feature 'C' 

     START(xP1,yP2) 

     LINE TO(xP2,yP2) 

Feature 'H' 

     START(xP6,yP6) 

     LINE TO(xP5,yP5) 

PLOTS 

   GRID(x,y) 

   CONTOUR(Temp)  notags nominmax 

   REPORT FluxC AS "FluxC" 

   REPORT FluxH AS "FluxH" 

   REPORT (FluxC-FluxH)/(0.5*(FluxC+FluxH)) AS "delFlux" 

   SURFACE(Temp) VIEWPOINT(-10,-1000000,90) 

   VECTOR(grad(Temp)) 

END 

 

Результаты моделирования, полученные при указанных в тексте 

программы значениях параметров приведены на следующих рисунках. 

 

Рис. 3.2. 

Рис. 3.3. 

 

 
Рис. 3.4. 

Рис. 3.5. 

 

Значения переменных FluxH и FluxC (см. рис. 3.3) соответствуют потоку 

тепла через верхнюю и нижнюю границы, соответственно. При точном 

решении задачи величины этих потоков должны быть равны, так что 

разность их значений delFlux (см. рис. 3.3) является характеристикой 

погрешности численного решения. 

 
3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, объектом является 

стационарный теплоперенос в двумерной области. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Составить и ввести в систему FlexPDE текст программы, выполнить 

расчеты. Внести изменения в текст программы по задания преподавателя 
3. Подготовить отчет о выполнении работы. 
 
 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания программы решения задачи стационарного теплопереноса в среде 
FlexPDE, текст прикладной программы и полученные результаты 
моделирования. 

 
 
 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой вид имеет уравнение Лапласа? 

2. В какой геометрической области решается уравнение Лапласа в 

специализированной среде FlexPDE? 

3. Какие разделы имеет задание к FlexPDE? 

4. Чему по умолчанию равна относительная погрешность во FlexPDE? 

5. Как задается производная во FlexPDE?  

6. В каком виде получаем результат численного моделирования во 

FlexPDE? 
 
 

7. ИСПОЛЬЗУЕМАЯ ЛИТЕРАТУРА 
1. Дворецкий С.И., Ермаков А.А., Иванов О.О., Акулинин Е.И. 

Компьютерное моделирование процессов и аппаратов пищевой, био- и 

химической технологии в среде FlexPDE: Учебное пособие. - Тамбов: 

Издательство ТГТУ, 2006. - 72 с.    http://window.edu.ru/resource/586/38586 

2. Маслов Ю.А., Меринов И.Г., Рябов Н.О. Моделирование 

теплогидравлических процессов в реакторных установках и элементах 

теплообменного оборудования ЯЭУ: лабораторный практикум. – М.: МИФИ, 

2008. – 156 с.   http://www.studfiles.ru/preview/413352/page:8/ 

3. FlexPDE 6: User Guide-

http://www.pdesolutions.com/help/index.html?user_guide.html 

4. FlexPDE - численное решение систем дифференциальных уравнений 

в частных производных - http://www.1024.ru/studentam/flexpde 

5. Математическое моделирование конвективного тепломассообмена на 

основе уравнений Навье-Стокса / В.И. Полежаев, А.В. Бунэ, Н.А. Верезуб и 

др. М.: Наука. 1987. 

6. Ши Д. Численные методы в задачах теплообмена. М.: Мир. 1988. 
 

http://window.edu.ru/resource/586/38586
http://www.studfiles.ru/preview/413352/page:8/
http://www.pdesolutions.com/help/index.html?user_guide.html
http://www.1024.ru/studentam/flexpde


32 

ЛАБОРАТОРНАЯ РАБОТА № 4 

Моделирование нестационарного теплопереноса 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач нестационарного теплопереноса в 

среде FlexPDE, исследовать влияние параметров системы на распределение 

температуры и величину теплового потока, закрепить теоретические знания, 

полученные при изучении дисциплины. 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Ранее были рассмотрены стационарные режимы теплообмена, т. е. 

такие, в которых температурное поле по времени не изменяется и в 

дифференциальном уравнении теплопроводности Фурье — Кирхгофа 

производная ∂T / ∂τ = 0 . 

Однако целый ряд важных практических задач теплообмена не может 

быть рассмотрен в рамках предположения о неизменности параметров 

процесса по времени. К ним относятся задачи о прогреве теплозащитных 

оболочек и конструктивных элементов скоростных летательных аппаратов, о 

нагреве стенок сопел реактивных двигателей твердого топлива, о расчете 

поля температур в энергетических ядерных реакторах при изменении режима 

работы, о тепловом режиме искусственного спутника Земли (ИСЗ). В этой 

главе будут рассмотрены нестационарные процессы теплопроводности в 

неподвижных средах (твердых телах) и даны аналитические и численные 

методы решения дифференциального уравнения Фурье — Кирхгофа для 

нестационарного случая с различными краевыми условиями. 

Нестационарные тепловые процессы сопровождаются не только 

изменением температурного поля по времени, но почти всегда связаны с 

изменением энтальпии тела, т. е. с его нагревом и охлаждением. 

Практические задачи нестационарного теплообмена можно разделить 

на две основные группы. К первой относятся процессы, происходящие при 

переходе тепла из некоторого начального теплового состояния в иное 

стационарное, обычно равновесное тепловое состояние. Примерами могут 

служить изменение температурного поля в теле, помещенном в среду, 

температура которой отличается от начальной температуры тела, или 

выравнивание температур в теле с заданным начальным распределением 

температур. Ко второй группе можно отнести процессы, происходящие в 

телах, испытывающих тепловое воздействие извне, изменяющиеся во 

времени по некоторому закону. Здесь можно назвать процессы 

периодического изменения температуры при движении ИСЗ по орбите, часть 

которой пролегает в тени Земли, суточные и годовые колебания температуры 

в верхних слоях земной коры, тепловые режимы аппаратов, находящихся на 

поверхности Луны, процессы в регенеративных теплообменниках и др. 

В большинстве нестационарных тепловых процессов можно выделить 

три этапа, характеризующиеся различными режимами, из которых 

собственно нестационарными будут лишь два первых. На первом этапе поле 

температур в теле определяется не только изменившимся тепловым 

воздействием, например изменением температуры окружающей среды, но и 

начальным распределением температур в теле T0(x, y, z) при τ = 0. Поскольку 

начальное температурное поле в общем случае может быть весьма 

произвольным, то и тепловой режим на этом первом этапе носит характер 

неупорядоченного процесса. 

На втором этапе влияние начального состояния все более и более 

ослабевает, и дальнейшее протекание процесса управляется лишь условиями 

на границе тела, т. е. наступает режим упорядоченного процесса, в частности, 

регулярный режим. 

Для большинства процессов первой группы характерен еще и третий 

этап, в котором температура тела во всех точках одинакова и равна 

температуре окружающей среды. Это состояние называют состоянием 

теплового равновесия. 

Строго говоря, это новое равновесное тепловое состояние наступает 

лишь по прошествии бесконечно большого промежутка времени. Однако на 

практике тело относительно быстро достигает состояния, весьма близкого к 

состоянию теплового равновесия, поэтому и интересующие нас длительности 

нестационарных режимов отнюдь не бесконечны. 

Дифференциальное уравнение нестационарной теплопроводности в 

случае неподвижной среды и отсутствия внутренних источников тепла имеет 

вид: 

где с, ρ – удельная теплоемкость и плотность среды, соответственно; λ – 

коэффициент теплопроводности; t – время. 

Если свойства среды являются постоянными, то уравнение (4.1) в 

декартовых координатах можно записать в следующем виде: 

где a= λ/(сρ) – коэффициент температуропроводности. 

При анализе нестационарного теплопереноса в двумерной области (рис. 

4.2) будем считать, что участки границы, обозначенные нижним индексом И 

(ГИ1, ГИ2, ГИ3 и ГИ4) являются теплоизолированными, то есть тепловой поток 

через эти границы равен нулю: 

где n – единичный вектор внешней нормали к соответствующему участку 

границы области. 

Верхняя (ГН) и нижняя (ГС) границы являются изотермическими, то есть 

на них поддерживается постоянная температура: 

CH
CH

  , TTTT 
  (4.4) 

Кроме того, примем, что в начальный момент времени температура в 

расчетной области Ω была равна TC: 

C0,
TT

t


  (4.5) 

 

 
 

Рис. 4.1. 

В результате численного решения задачи требуется определить как 

будет изменяться распределение температур в рассматриваемой области с 

течением времени и определить величины тепловых потоков через верхнюю 

и нижнюю границы области. 

Для численного решения рассматриваемой задачи в среде FlexPDE 

может быть использована следующая программа 
TITLE NonSteady State 2D Heat Transfer 
COORDINATES 
  cartesian2 
VARIABLES 
Temp 
DEFINITIONS 
A=1 
Tmod=20 
dTmod=0.5 
L=5        L1=2 
H=2       H1=4 
xP1=0    yP1=0 

xP2=L   yP2=0 
xP3=L   yP3=H 
xP4=L1 yP4=H 
xP5=L1 yP5=H1 
xP6=0 yP6=H1 
TempH=10  TempC=0 
FluxC=LINE_INTEGRAL(normal(grad(Temp)),'C') 
FluxH=-LINE_INTEGRAL(normal(grad(Temp)),'H') 
INITIAL VALUE 
  Temp = TempC 
EQUATIONS 
Temp: dt(Temp)=A*Div(grad(Temp)) 
BOUNDARIES 
REGION 1 
    START(xP1,yP1) 
      VALUE(Temp)=TempC 
     LINE TO(xP2,yP2) 
      NATURAL(Temp)=0 
    LINE TO (xP3,yP3) 
      NATURAL(Temp)=0 
    LINE TO (xP4,yP4) 
      NATURAL(Temp)=0 
    LINE TO (xP5,yP5) 
      VALUE(Temp)=TempH 
    LINE TO (xP6,yP6) 
      NATURAL(Temp)=0 
    LINE TO CLOSE  
Feature 'C' 
     START(xP1,yP2) 
     LINE TO(xP2,yP2) 
Feature 'H' 
     START(xP6,yP6) 
     LINE TO(xP5,yP5) 
TIME 0 TO Tmod   
PLOTS 
  for t = 0 by dTmod to Tmod 
   GRID(x,y) 
   CONTOUR(Temp)  notags nominmax 
   REPORT FluxC AS "FluxC" 
   REPORT FluxH AS "FluxH" 
   REPORT (FluxC-FluxH)/(0.5*(FluxC+FluxH)) AS "delFlux" 
   SURFACE(Temp) VIEWPOINT(-10,-1000000,90) 
   VECTOR(grad(Temp)) 
   history(FluxC) 
END 

Результаты моделирования, полученные при указанных в тексте 

программы значениях параметров приведены на следующих рисунках. 

 

Рис. 4.2. 

 

   

  
 

Рис. 4.3. 

t=0.25 t=0.5 

t=1 t=5 

На рис. 4.4 приведено изменение величины теплового потока через 

нижнюю границу области с течением времени 

 

 
Рис. 4.4. 

 

Значения переменных FluxH и FluxC  соответствуют потоку тепла через 

верхнюю и нижнюю границы, соответственно. При точном решении задачи 

величины этих потоков должны быть равны, так что разность их значений 

delFlux является характеристикой погрешности численного решения. 

 
3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, объектом является 

нестационарный теплоперенос в двумерной области. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Составить и ввести в систему FlexPDE текст программы, выполнить 

расчеты. Внести изменения в текст программы по задания преподавателя 
3. Подготовить отчет о выполнении работы. 
 
 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания программы решения задачи нестационарного теплопереноса в 

среде FlexPDE, текст прикладной программы и полученные результаты 
моделирования. 

 
 
 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой вид имеет уравнение нестационарной теплопроводности в 

неподвижной среде? 

2. В какой геометрической области решается уравнение нестационарной 

теплопроводности в специализированной среде FlexPDE? 

3. Чему по умолчанию равна относительная погрешность во FlexPDE? 

4. Как задается производная по времени в системе FlexPDE?  

5. В каком виде получаем результат численного моделирования во 

FlexPDE? 
 
 
 

7. ИСПОЛЬЗУЕМАЯ ЛИТЕРАТУРА 
1. Дворецкий С.И., Ермаков А.А., Иванов О.О., Акулинин Е.И. 

Компьютерное моделирование процессов и аппаратов пищевой, био- и 

химической технологии в среде FlexPDE: Учебное пособие. - Тамбов: 

Издательство ТГТУ, 2006. - 72 с.    http://window.edu.ru/resource/586/38586 

2. Маслов Ю.А., Меринов И.Г., Рябов Н.О. Моделирование 

теплогидравлических процессов в реакторных установках и элементах 

теплообменного оборудования ЯЭУ: лабораторный практикум. – М.: МИФИ, 

2008. – 156 с.   http://www.studfiles.ru/preview/413352/page:8/ 

3. FlexPDE 6: User Guide-

http://www.pdesolutions.com/help/index.html?user_guide.html 

4. FlexPDE - численное решение систем дифференциальных уравнений 

в частных производных - http://www.1024.ru/studentam/flexpde 

5. Математическое моделирование конвективного тепломассообмена на 

основе уравнений Навье-Стокса / В.И. Полежаев, А.В. Бунэ, Н.А. Верезуб и 

др. М.: Наука. 1987. 

6. Ши Д. Численные методы в задачах теплообмена. М.: Мир. 1988. 

http://window.edu.ru/resource/586/38586
http://www.studfiles.ru/preview/413352/page:8/
http://www.pdesolutions.com/help/index.html?user_guide.html
http://www.1024.ru/studentam/flexpde

ЛАБОРАТОРНАЯ РАБОТА № 5 

Моделирование течения жидкости при обтекании твердых тел 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач течения несжимаемой вязкой 

жидкости при обтекании тел, исследовать влияние параметров системы на 

характер течения жидкости и величину силы сопротивления, закрепить 

теоретические знания, полученные при изучении дисциплины. 

 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Движение несжимаемой вязкой жидкости при обтекании твердого тела 

описывается уравнениями Навье-Стокса: 

где V – вектор гидродинамической скорости; ρ – плотность жидкости; p – 

давление; ν – коэффициент кинематической вязкости. 

В декартовых координатах система уравнений (5.1) может быть 

записана в следующем виде: 

 (5.2) 

К сожалению, последнее уравнение в (5.2) не может быть использовано 

в среде FlexPDE. Используя уравнения движения и уравнение неразрывности 

это уравнение может быть заменено дифференциальным уравнением второго 

порядка для давления, которое имеет следующий вид: 

  VV  p  (5.3) 

Или в декартовых координатах: 

Хотя уравнение (5.3) было получено с учетом уравнения 

неразрывности не любое решение этого уравнения будет удовлетворять 

последнему уравнению системы (5.2). Для гарантированного выполнения 

условия несжимаемости жидкости следует добавить к уравнению (5.3) 

дополнительный член - 

V

yx , где f  - коэффициент, определяемый 

следующим соотношением: 

  (5.5) 

где L – характерный размер (например, диаметр обтекаемого тела); С – 

эмпирический коэффициент, значение которого должно быть достаточно 

большим, чтобы гарантировать выполнение условия неразрывности, но не 

очень большим, чтобы обеспечить сходимость численного решения. 

Таким образом, для моделирования обтекания твердого тела потоком 

несжимаемой вязкой жидкости будем использовать следующую систему 

дифференциальных уравнений: 

На поверхности стенок и твердого тела, находящегося в потока, 

скорость жидкости равна нулю. На большом удалении от обтекаемого 

препятствия отлична от нуля только горизонтальная составляющая скорости. 

В отличие от гидродинамической скорости, для которой граничные условия 

на границах расчетной области известны, граничные условия для давления 

заранее не заданы и должны быть получены из уравнений движения и 

условия неразрывности. 

Из первого уравнения (5.6) для давления получим следующее 

соотношение: 

где n – единичный вектор нормали к границе расчетной области. 

Так как на поверхности твердого тела гидродинамическая скорость 

равна нулю, то из соотношения (5.7) окончательно получим следующее 

соотношение, выражающее граничное условие для давления: 

Vn 





n

p
 (5.8) 

Рассмотрим задачу об обтекании цилиндра, расположенного в канале 

(геометрия расчетной области и сетка конечных элементов приведена на рис. 

5.1). 

Рис. 5.1. 

 

Для выполнения расчетов будем использовать следующую программу: 

TITLE 'Viscous Flow past a Circular Cylinder' { fex211.pde } 

SELECT errlim=5e-3 ngrid=1 spectral_colors stages=2 

VARIABLES vx vy p 

DEFINITIONS 

Lx=2.0 Ly=1.0 a=0.2 visc=1e4 

delp=100 { Driving pressure } 

dens=1e3 Re=dens*globalmax( vx)*2*Lx/visc 

v=vector( vx, vy) vm=magnitude( v) 

unit_x=vector(1,0) unit_y=vector(0,1) { Unit vector fields } 

nx=normal( unit_x) ny=normal( unit_y) { Direction cosines } 

natp= 

if stage=2 then visc*[ nx*div( grad( vx))+ ny*div( grad( vy))] else 0 

EQUATIONS 

vx: dx( p)- visc*div( grad( vx))=0 

vy: dy( p)- visc*div( grad( vy))=0 

p: div( grad( p))- 1e4*visc/Ly^2*div( v)=0 

BOUNDARIES 

region 'domain' start 'outer' (-Lx,Ly) 

natural( vx)=0 natural( vy)=0 value(p)=delp { In } 

line to (-Lx,-Ly) value( vx)=0 value( vy)=0 natural(p)=natp { Wall} 

line to (Lx,-Ly) natural( vx)=0 natural( vy)=0 value(p)=0 { Out } 

line to (Lx,Ly) value( vx)=0 value( vy)=0 natural(p)=natp { Wall } 

line to close 

start 'outline' (a,0) { Exclude cylinder } 

value( vx)=0 value( vy)=0 natural(p)=natp 

arc( center=0,0) angle=360 close 

PLOTS 

contour( vx) report( Re) contour( vy) 

contour( vm) painted contour( p) 

vector( v) norm vector(v) norm zoom(-2*a,-2*a, 4*a,4*a) 

contour( div( v)) contour( curl( v)) painted 

elevation( vx, vy) from (-Lx,0) to (Lx,0) 

END 

 

Результаты расчетов представлены на рис. 5.2 – 5.6. 

 
Рис. 5.2. 

Рис. 5.3. 

 

 
Рис. 5.4. 

Рис. 5.5. 

 

 
Рис. 5.6. 

 
3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, объектом является 

течение несжимаемой вязкой жидкости при обтекании твердого тела. 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Выполнить текстовые расчеты по программам указанным 

преподавателем. 
3. Ввести в систему FlexPDE текст программы, приведенный в 

теоретической части, и выполнить расчеты. Внести изменения в текст 
программы по задания преподавателя 

4. Подготовить отчет о выполнении работы. 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания основных разделов описания задач в среде FlexPDE, текст 
прикладной программы и полученные результаты моделирования. 

 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой физический смысл имеют уравнения движения несжимаеморй 

вязкой жидкости? 
2. Какое уравнение используется в системе FlexPDE для расчета 

давления? 
3. Какие граничные условия для скорости жидкости задаются на 

поверхности твердого тела? 
4. Поясните полученный характер движения жидкости при обтекании 

цилиндра? 
 

7. ИСПОЛЬЗУЕМАЯ ЛИТЕРАТУРА 
1. Дворецкий С.И., Ермаков А.А., Иванов О.О., Акулинин Е.И. 

Компьютерное моделирование процессов и аппаратов пищевой, био- и 

химической технологии в среде FlexPDE: Учебное пособие. - Тамбов: 

Издательство ТГТУ, 2006. - 72 с.    http://window.edu.ru/resource/586/38586 

2. Маслов Ю.А., Меринов И.Г., Рябов Н.О. Моделирование 

теплогидравлических процессов в реакторных установках и элементах 

теплообменного оборудования ЯЭУ: лабораторный практикум. – М.: МИФИ, 

2008. – 156 с.   http://www.studfiles.ru/preview/413352/page:8/ 

3. FlexPDE 6: User Guide-

http://www.pdesolutions.com/help/index.html?user_guide.html 

4. FlexPDE - численное решение систем дифференциальных уравнений 

в частных производных - http://www.1024.ru/studentam/flexpde 

5. Математическое моделирование конвективного тепломассообмена на 

основе уравнений Навье-Стокса / В.И. Полежаев, А.В. Бунэ, Н.А. Верезуб и 

др. М.: Наука. 1987. 

6. Слеттери Дж.С. Теория переноса импульса, энергии и массы в 

сплошных средах. М.: Энергия. 1978. 

7. Патанкар С. Численные методы решения задач теплообмена и 

динамики жидкости. М.: Энергоатомиздат. 1984. 

http://window.edu.ru/resource/586/38586
http://www.studfiles.ru/preview/413352/page:8/
http://www.pdesolutions.com/help/index.html?user_guide.html
http://www.1024.ru/studentam/flexpde

ЛАБОРАТОРНАЯ РАБОТА № 6 

Моделирование естественной конвекции неравномерно нагретой 

жидкости 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач течения несжимаемой вязкой 

неравномерно нагретой жидкости, исследовать влияние параметров системы 

на характер течения жидкости и величину потока тепла через границу 

области, закрепить теоретические знания, полученные при изучении 

дисциплины. 

 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

В приближении Буссинеска уравнения движения несжимаемой вязкой 

жидкости и уравнения нестационарного конвективного теплопереноса имеют 

следующий вид: 

где V – вектор гидродинамической скорости; ρ0 – плотность жидкости при 

температуре T0; p – давление; ν – коэффициент кинематической вязкости; T – 

температура; a – коэффициент температуропроводности; F – выталкивающая 

(Архимедова) сила в поле земного притяжения, учитывающая зависимость 

плотности жидкости от температуры. 

В линейном приближении для плотности жидкости можно 

использовать следующее соотношение: 

  00 1 TT    (6.2) 

где 
T









0

1
 - температурный коэффициент, характеризующий изменение 

плотности жидкости от температуры, связанное с ее объемным расширением 

при повышении температуры. 

Учитывая (6.1) выражение для силы F можно записать так: 

   000 TTF   gg  (6.3) 

где g – ускорение свободного падения. 

В двумерном случае система уравнений (6.1) с учетом (6.3) может быть 

записана в декартовой системе координат следующем виде: 

 (6.4) 

Как было отмечено ранее (см. Лаб. Работу №5) уравнение 

неразрывности (третье уравнение в системе (6.4)) не может быть 

непосредственно решено в среде FlexPDE. В предыдущей лабораторной 

работе уравнение неразрывности было заменено дифференциальным 

уравнением второго порядка для давления. Для исключения уравнения 

неразрывности можно использовать и другие подходы. В настоящей 

лабораторной работе для исключения уравнения неразрывности используем 

метод, основанный на введении функции тока и завихренности. 

Введем функцию тока ψ, которая связана с компонентами 

гидродинамической скорости следующими соотношениями: 

x

y
V yx












   ,  (6.5) 

Подставляя соотношения (6.4) в уравнение неразрывности 

yxyxy

V yx 
, (6.6) 

видим, что уравнение неразрывности для функции тока тождественно 

выполняется и поэтому уравнение неразрывности при введении функции 

тока включать в математическую модель не требуется. 

Дифференцируя первое уравнение в системе (6.4) по y, а второе 

уравнение - по x, и вычитая из первого полученного соотношение второе, 

получим уравнение, из которого исключено давление: 

2  (6.7) 

Вводя завихренность, которая связана с функцией тока и компонентами 

гидродинамической скорости следующим соотношением: 

Подставляя соотношения (6.5) и (6.8) в (6.4) и (6.7) получим 

следующую систему уравнений для движения и теплопереноса в 

неравномерно нагретой жидкости: 

 (6.9) 

Учитывая, что на поверхностях твердых тел выполняется условие 

прилипания, т.е. скорость жидкости равна нулю, можно принять нулевое 

значение функции тока. Граничное условие 

0


 , (6.10) 

фактически выражает условие отсутствие потока жидкости через границу 

расчетной области, однако касательная составляющая скорости может быть 

отлична от нуля. Для обеспечения нулевого значения касательной 

составляющей скорости на границе требуется задать дополнительное 

граничное условие для завихренности. В качестве такого условия мы будем 

считать, что на поверхности действует источник завихренности 

направленный противоположно касательной составляющей 

гидродинамической скорости. Граничное условие для завихренности может 

быть записано в следующем виде: 

где С – константа, характеризующая интенсивность поверхностного 

источника завихренности. При увеличении значения C касательная 

составляющая скорости уменьшается, однако при этом ухудшается 

сходимость численного решения. Значение константы C следует выбирать 

таким образом, чтобы обеспечивалась удовлетворительная сходимость и 

касательная составляющая скорости имела малое значение, не оказывающее 

существенного слияния на движение жидкости и конвективный теплоперенос 

в условиях естественной конвекции. 

 

Для моделирования рассматриваемой задачи во FlexPDE можно 

использовать следующую программу 

 

TITLE 'Natural convection by Stream Function and Vorticity' 

COORDINATES CARTESIAN2 

VARIABLES 

   Temp(10)    psi    w(1e-4) 

DEFINITIONS 

 visc=1e-6 

 atemp=1e-6 

 galpha=1e-3 

 L=2e-1 

 H=1e-1 

 C=0 

 TC=0 

 TH=10 

!   MESH_DENSITY = 5 

   MESH_SPACING=0.01 

  u = -dy(psi) 

  v = dx(psi) 

  Sh=line_integral(abs(normal[grad(Temp)])/(L*TH),"s") 

! INITGRIDLIMIT =5 

SELECT 

  REGRID = Off 

  ERRLIM=0.001 

 NODELIMIT=3000 

INITIAL VALUE 

   Temp =TH+y/H*(TC-TH) 

!  Temp=0; 

   psi=0.00 

   w=0.00 

EQUATIONS 

 Temp:  dt(Temp) + u*dx(Temp) + v*dy(Temp)=atemp*div(grad(Temp)) 

  psi: div(grad(psi)) + w = 0 

  w: dt(w) + u*dx(w) + v*dy(w) =visc*div(grad(w))- galpha*dx(Temp)  

BOUNDARIES 

   region 1 

      start "s" (0,0) 

      value(Temp) = TH 

      value(psi) = 0 

      natural(w)=C*tangential(u,v) 

      line to  (L,0) 

      natural(Temp) = 0 

      value(psi) = 0 

      natural(w)=C*tangential(u,v) 

      line to  (L,H) 

      value(Temp) = TC 

      value(psi) = 0 

      natural(w)=C*tangential(u,v) 

      line to  (0,H) 

      natural(Temp) = 0 

      value(psi) = 0 

      natural(w)=C*tangential(u,v) 

     line to close 

TIME 0.001 to 200 

MONITORS 

   for cycle=1  

   contour(Temp) 

   contour(psi) as "Stream Function"  NOMINMAX 

   contour(w) as "Vorticity" 

   vector(curl(psi)) as "Flow Velocity" 

PLOTS 

   for t = 0.1 by 0.1 to endtime 

   grid(x,y) 

   contour(Temp)  contours=10 fixed range=(0.0,TH) as "Temp"  NOMINMAX 

   SURFACE(Temp) VIEWPOINT(-10,-1000000,90) 

   contour(psi) as "Stream Function" 

   contour(w) as "Vorticity" 

   vector(curl(psi)) as "Flow Velocity" norm 

    history (Sh) 

HISTORIES 

!    history (Sh, Sh0) as "Sh" export format "#t#r,#i"  file="Sh"+$Var+".dat" 

!    history (Vav) as "Vav" export format "#t#r,#i"  file="Vav"+$Var+".dat" 

END 

 

 

 

На рис. 6.1 приведена геометрия расчетной области и сетка конечных 

элементов. 

 
Рис. 6.1. 

 

На рис. 6.2 приведены результаты численного моделирования. 

                          t=0                                                     t=30 

  
                          t=35                                                     t=40 

 

Рис. 6.2. 

 
3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, конвективный 

теплоперенос в неравномерно нагретой вязкой жидкости. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Выполнить текстовые расчеты по программам указанным 

преподавателем. 
3. Ввести в систему FlexPDE текст программы, приведенный в 

теоретической части, и выполнить расчеты. Внести изменения в текст 
программы по задания преподавателя 

4. Подготовить отчет о выполнении работы. 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания основных разделов описания задач в среде FlexPDE, текст 
прикладной программы и полученные результаты моделирования. 

 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой физический смысл имеют уравнения конвективной 

теплопроводности? 
2. Как изменяется величина теплового потока через поверхность 

пластины по ее длине? 
3. С какой целью ввходятся новые переменные – функция тока и 

завихренность? 
4. При каком соотношении между температурами верхней и нижней 

границ расчетной области будет возникать конвективное движение 
жидкости? 

ЛАБОРАТОРНАЯ РАБОТА № 7 

Моделирование напряженного состояние деформируемого твердого тела 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач механики деформируемого 

твердого тела, исследовать влияние параметров системы на характер 

распределения напряжений и перемещений в пластине с отверстием, 

закрепить теоретические знания, полученные при изучении дисциплины. 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Рассмотрим деформацию пластины с отверстием под действием 

растягивающих напряжений. Уравнения для напряжений/деформаций, 

являющиеся следствием баланса сил в материале пластины, могут быть 

записаны в следующем виде: 

 

dx(Sx) + dy(Txy) + Fx = 0 

dx(Txy) + dy(Sy) + Fy = 0 

 

где Sx и Sy - напряжения вдоль координатных осей x и y, соответственно; 

Txy – касательное напряжение (напряжение сдвига); Fx и Fy – объемные 

силы в напрялениях координатных осей x и y, соответственно. 

Деформация материала описывается перемещениями U и V вдоль 

координатных осей x и y, которые связаны с деформациями посредством 

следующих соотношений 

 

ex = dx(U) 

ey = dy(V) 

gxy = dy(U) + dx(V). 

 

Восемь величин U, V, ex, ey, gxy, Sx, Sy и Txy связаны между собой 

определяющими соотношениями. В общем случае эти соотношения можно 

записать в следующем виде: 

 

Sx  =  C11*ex + C12*ey + C13*gxy - b*Temp 

Sy  =  C12*ex + C22*ey + C23*gxy - b*Temp 

Txy =  C13*ex + C23*ey + C33*gxy 

 

В изотропном твердом теле C13 = C23 = 0. 

Учитывая эти соотношения, мы получим уравнения для перемещений: 

 

dx[C11*dx(U)+C12*dy(V)] + dy[C33*(dy(U)+dx(V))] + Fx = dx(b*Temp) 

dy[C12*dx(U)+C22*dy(V)] + dx[C33*(dy(U)+dx(V))] + Fy = dy(b*Temp) 

В приближении плоского напряженного состояния, которое допустимо 

использовать для тонких и длинных пластин, можно использовать 

следующие соотношения: 

 

C11 = G 

C12 = G*nu 

C22 = G 

C33 = G*(1-nu)/2 

 

где G = E/(1-nu**2); E – модуль Юнга; nu – коэффициент Пуассона. 

 

Уравнения для определения перемещений можно записать в 

следующем виде (при постоянной температуре и при отсутствии объемных 

сил): 

 

dx[dx(U)+nu*dy(V)] + 0.5*(1-nu)*dy[dy(U)+dx(V)] = 0 

dy[nu*dx(U)+dy(V)] + 0.5*(1-nu)*dx[dy(U)+dx(V)] = 0 

 

Для задания граничных условий, рассмотрим уравнения для 

напряжений в их начальном виде: 

 

dx(Sx) + dy(Txy) = 0 

dx(Txy) + dy(Sy) = 0 

 

Эти уравнения можно записать и так 

 

div(P) = 0 

div(Q) = 0 

 

где P = [Sx,Txy]    and   Q = [Txy,Sy] 

 

Граничные условия типа "load" (или "natural") для U определяют 

составляющую P в направлении внешней нормали к границе, аналогичные 

граничные условия для V определяют нормальную составляющую Q. Таким 

образом, граничные условия типа “load” для  U и V совместно определяют 

вектор напрузки на поверхности. 

На свободной поверхности вектор нагрузки равен нулю, поэтому 

граничные условия могут быть записаны следующим образом: 

 

load(U) = 0 

load(V) = 0. 

 

Рассмотрим пластину с отверстием, к которой приложена 

растягивающая нагрузка в направлении оси X. Ниже приведена программа 

для расчета напряженного состояния этой пластины. 

 

  Title 'Плоское напряженное состояние в пластине с отверстием' 

  Select 

    errlim = 1e-4 { увеличение точности расчета напряжений} 

    painted { применять заливку для всех контурных графиков } 

  Variables 

    U                   {  U и V переменные} 

    V 

  Definitions 

    nu = 0.3            { коэффициент Пуассона } 

    E  = 21             { модуль Юнга x 10**-11 } 

    G  = E/(1-nu**2) 

    C11 = G 

    C12 = G*nu 

    C22 = G 

    C33 = G*(1-nu)/2 

    p1 = (1-nu)/2       { = C33/G } 

  Initial values 

    U = 1               { начальные значения переменных } 

    V = 1 

  Equations             { уравнения для плоского напряженного состояния } 

     U: dx[dx(U) + nu*dy(V)] + p1*dy[dy(U) + dx(V)]  = 0 

     V: dy[dy(V) + nu*dx(U)] + p1*dx[dy(U) + dx(V)]  = 0 

  

  Boundaries 

    region 1 

      start (0,0) 

      load(U)=0      { свободная граница, отсутствуют нормальные напряжения } 

      load(V)=0 

      line to (3,0)      

      load(U)=0.1    { задание напряжений вдоль оси X на правой границе} 

      load(V) = 0 

      line to (3,1) 

      load(U)=0      { свободная верхняя граница } 

      load(V)=0 

      line to (0,1) 

      value(U)=0        { условие жесткой заделки на левой границе } 

      value(V)=0 

      line to close 

                        { Задание отверстия в пластине } 

      load(U) = 0 

      load(V) = 0 

      start(1,0.25) 

      arc(center=1,0.5) angle=-360 

  Monitors 

    grid(x+U,y+V) {отображение деформированной сетки в процессе решения} 

  Plots  
     grid(x+U,y+V)      { вывод конечной деформированной сетки } 

     vector(U,V) as "Displacement"        { вывод смещений } 

     contour(U) as "X-Displacement" 

     contour(V) as "Y-Displacement" 

     contour((C11*dx(U) + C12*dy(V))) as "X-Stress" 

     contour((C12*dx(U) + C22*dy(V))) as "Y-Stress" 

     surface((C11*dx(U) + C12*dy(V))) as "X-Stress" 

     surface((C12*dx(U) + C22*dy(V))) as "Y-Stress" 

  End 

На рис. 7.1 приведены исходная (а) и деформированная (б) форма 

пластины с отверстием. 

   

а б 

Рис. 7.1. 

На рис. 7.2 представлены смежения по оси Х (а) и по оси У (б), 

соответственно. 

  

а б 

Рис. 7.2. 

На рис. 7.3 представлены напряжения по оси Х (а) и по оси У (б), 

соответственно. 

  

а б 

Рис. 7.3. 

3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, напряженное состояние 

пластины с отверстием. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Выполнить текстовые расчеты по программам указанным 

преподавателем. 
3. Ввести в систему FlexPDE текст программы, приведенный в 

теоретической части, и выполнить расчеты. Внести изменения в текст 
программы по задания преподавателя 

4. Подготовить отчет о выполнении работы. 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания основных разделов описания задач в среде FlexPDE, текст 
прикладной программы и полученные результаты моделирования. 

 
 
 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой физический смысл имеют уравнения для напряжений? 
2. Какое приближение используется для расчета напряженного 

состояния пластины с отверстием? 
3. Как связаны между собой перемещения и дефолрмации? 
4. Как по результатам расчета найти коэффициент концентрации 

напряжений? 

ЛАБОРАТОРНАЯ РАБОТА № 8 

Моделирование механических вибраций деформируемого твердого тела 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач механики деформируемого 

твердого тела при периодически изменяющейся нагрузке, исследовать 

влияние параметров системы на характер распределения напряжений и 

перемещений в стержне, совершающем гармонические колебания, закрепить 

теоретические знания, полученные при изучении дисциплины. 

 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Рассмотрим пример использования системы FlexPDE для анализа 

механичнеских колебаний деформируемого твкердого тела. 

Уравнения для напряжений / деформаций в деформируемом твердом 

теле можно записать в следующем виде: 

 

dx(Sx) + dy(Txy) + Fx = 0 

dx(Txy) + dy(Sy) + Fy = 0 

 

где Sx и Sy – напряжения в напралениях x и y, соответственно; Txy – 

касательное напряжение (напряжение сдвига); Fx и Fy – составляющие 

объемной силы в направлении координатных осей x и y, соответственно. 

В нестационарном случае ускорение материала и вязкие силы 

выступают в качестве объемных сил и поэтому должны быть включены в 

следующие соотношения: 

 

Fx1 = Fx0 - rho*dtt(U) + mu*del2(dt(U)) 

Fy1 = Fy0 - rho*dtt(V) + mu*del2(dt(V)) 

 

где rho – плотность материала; mu – вязкость материала; U и V – 

перемещения по координатным осям x и y, соответственно. 

Будем считать, что перемещения периодически изменяются с течением 

времени, тогда можно записать: 

 

U(t) = U0*exp(-i*omega*t) 

V(t) = V0*exp(-i*omega*t) 

 

Здесь U0(x,y) and V0(x,y) – комплексные амплитуды перемещений; 

omega – угловая скорость is колебаний (вибраций). 

Подставляя эти соотношения в уравнения для напряжений, после 

выполнения ряда преобразований, получим 

 

dx(Sx) + dy(Txy) + Fx0 + rho*omega^2*U - i*omega*mu*del2(U) = 0 

dx(Txy) + dy(Sy) + Fy0 + rho*omega^2*V - i*omega*mu*del2(V) = 0 

Все члены в полученных уравнениях являются комплексными. 

Разделим действительные и мнимые части: 

 

U = Ur + i*Ui 

Sx = Srx + i*Six 

Sy = Sry + i*Siy 

etc... 

 

Определяющие соотношения можно записать следующим образом: 

 

Srx = [C11*dx(Ur) + C12*dy(Vr)] 

Sry = [C12*dx(Ur) + C22*dy(Vr)] 

Trxy = C33*[dy(Ur) + dx(Vr)] 

etc... 

 

В результате получим систему из четырех уравнений для Ur, Vr, Ui и 

Vi. 

Рассматрим задачу колебаний алюминиевой пластины, имеющей длину 

1 м и толщину 5 см, которая закреплена слева, а на ее правом конце 

приложена периодическая нагрузка. Частота изменения нагрузки близка к 

собственной частоте пластины. 

При выичсления мы рассмотрим следующие случаи: без учета 

вязкости, с учетом вязкости. 

 

  Title "Harmonic Stress analysis" 

  Variables 
    Ur   { Displacements } 

    Ui 

    Vr 

    Vi 

  Definitions 

    L = 1               { the bar length, in Meters } 

    hL = L/2 

    W = 0.05            { the bar thickness, in Meters } 

    hW = W/2 

    eps = 0.01*L 

    nu = 0.3            { Poisson's Ratio } 

    E  = 6.7e+10        { Young's Modulus for Aluminum (N/M^2) } 

                        { plane strain coefficients } 

    E1  = E/((1+nu)*(1-2*nu)) 

    C11 = E1*(1-nu) 

    C12 = E1*nu 

    C22 = E1*(1-nu) 

    C33 = E1*(1-2*nu)/2 

    rho = 2700          { Kg/M^3 } 

    mu = staged(0,1e3,1e4)    { Estimated viscosity Kg/M/sec } 

    cvel = sqrt(E/rho)  { sound velocity, M/sec } 

    tau  = L/cvel       { transit time } 

    tone = 0.25/tau     { approximate resonant frequency } 

    omega = 2*pi*tone       { driving angular velocity  } 

    amplitude=1e-8      { a guess for plot scaling } 

    mag=1/amplitude 

    force = 25          { loading force in Newtons (~1 pound force) } 

                        { distribute the force  uniformly over the driven end: } 

    fdist = force/W 

    Um = sqrt(Ur^2+Ui^2)        { X-displacement amplitude } 

    Vm = sqrt(Vr^2+Vi^2)        { X-displacement amplitude } 

    Srx = [C11*dx(Ur) + C12*dy(Vr)]        { Real Stresses } 

    Sry = [C12*dx(Ur) + C22*dy(Vr)] 

    Trxy = C33*[dy(Ur) + dx(Vr)] 

    Six = [C11*dx(Ui) + C12*dy(Vi)]        { Imaginary Stresses } 

    Siy = [C12*dx(Ui) + C22*dy(Vi)] 

    Tixy = C33*[dy(Ui) + dx(Vi)] 

    Sxm = sqrt(Srx^2+Six^2) 

    Sym = sqrt(Sry^2+Siy^2) 

    Txym = sqrt(Trxy^2+Tixy^2) 

  Equations             { define the displacement equations } 

    Ur: dx(Srx) + dy(Trxy) + rho*omega^2*Ur + omega*mu*del2(Ui) = 0 

    Ui: dx(Six) + dy(Tixy) + rho*omega^2*Ui - omega*mu*del2(Ur) = 0 

    Vr: dx(Trxy) + dy(Sry) + rho*omega^2*Vr + omega*mu*del2(Vi) = 0 

    Vi: dx(Tixy) + dy(Siy) + rho*omega^2*Vi - omega*mu*del2(Vr) = 0 

  Boundaries 

    region 1 

      start (0,-hW) 

      load(Ur)=0         { free boundary on bottom, no normal stress } 

      load(Ui)=0 

      load(Vr)=0 

      load(Vi)=0 

      line to (L,-hW) 

      load(Vr) = force   { Apply oscillatory vertical load on end. } 

      line to (L,hW) 

      load(Vr)=0         { free boundary on top, no normal stress } 

      line to (0,hW) 

      value(Ur) = 0      { clamp the left end  } 

      value(Ui) = 0 

      value(Vr) = 0 

      value(Vi) = 0 

      line to close 

  Monitors 

    elevation(Vr,Vi) from(0,0) to (L,0) 

      report(omega) report(mu) 

  Plots 

    grid(x+mag*Ur,y+mag*Vr) as "Real displacement" {show final deformed grid} 

      report(omega) report(mu) 

    grid(x+mag*Ui,y+mag*Vi)   as "Imag displacement" 

      report(omega) report(mu) 

    elevation(Vr,Vi) from(0,0) to (L,0) 

      report(omega) report(mu) 

    contour(Ur) as "Real X-displacement(M)" 

      report(omega) report(mu) 

    contour(Vr) as "Real Y-displacement(M)" 

      report(omega) report(mu) 

    contour(Ui) as "Imag X-displacement(M)" 

      report(omega) report(mu) 

    contour(Vi) as "Imag Y-displacement(M)" 

      report(omega) report(mu) 

    contour(Sxm) as "X-Stress amplitude" 

      report(omega) report(mu) 

    contour(Sym) as "Y-Stress amplitude" 

      report(omega) report(mu) 

    contour(Txym) as "Shear Stress amplitude" 

      report(omega) report(mu) 

  End 

 

На рис. 8.1 – 8.4 приведена геометрия расчетной области и схема 

конечных элементов (рис. 8.1) и результаты вычислительных экспериментов 

(рис. 8.2 – 8.5). 

 

Рис. 8.1. 

На рис. 8.2 представлены действительная (а) и мнимая (б) 

составляющие смещения. 

  

а б 

Рис. 8.2. 

На рис. 8.3 представлены действительная (а) и мнимая (б) 

составляющие скоростей смещения. 

 

Рис. 8.3. 

 

  

а 

 

б 

 

в 

Рис. 8.4. 

 

3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, объектом исследования 

является напряженное состояние стержня, совершающего гармонические 

колебания. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Выполнить текстовые расчеты по программам указанным 

преподавателем. 
3. Ввести в систему FlexPDE текст программы, приведенный в 

теоретической части, и выполнить расчеты. Внести изменения в текст 
программы по задания преподавателя 

4. Подготовить отчет о выполнении работы. 
 
 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания основных разделов описания задач в среде FlexPDE, текст 
прикладной программы и полученные результаты моделирования. 

 
 
 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Какой физический смысл имеют уравнения для напряжений? 
2. Какие силы должны учитываться в качестве объемных сил в случае 

нестационарного нагружения? 
3. Как связаны между собой перемещения и деформации? 
4. Как влияет вязкость материала на механические колебания 

деформируемой пластины? 

ЛАБОРАТОРНАЯ РАБОТА № 9 

Моделирование термических напряжений в деформируемом  

твердом теле 

 

1. ЦЕЛЬ РАБОТЫ 

Изучить особенности описания задач механики деформируемого 

твердого тела с учетом деформацией, обусловленных неравномерным 

распределением температуры, исследовать влияние параметров системы на 

характер распределения напряжений и перемещений в биметаллической 

пластине, ынутренняя часть которой нагревается за счет протекания 

электрического тока, закрепить теоретические знания, полученные при 

изучении дисциплины. 

2. ТЕОРЕТИЧЕСКИЕ ПОЛОЖЕНИЯ 

Ранее были рассмотрены случаи деформирования твердого тела, 

имеющего постоянную температуру. В данной работе будет исследовано 

влияние неравномерного распределения температуры на напряженное 

состояние тела. 

Стационарное распределение температуры в твердом теле описывается 

уравнением Пуассона: 

 
 

где f x и fy - составляющие вектора плотности теплового потока f, 

который характеризует скорость переноса тепла через единицу площади за 

единицу времени. Мощность внутреннего теплового источника h 

характеризует интенсивность выделения тепла в единичном объеме за 

единицу времени. Плотность теплового потока может быть представлена в 

следующем виде: 

 

 
 

где λ – коэффициент теплопроводности, T - темрература. 

Внутренне тепловыделение может быть связано, например, с 

протеканием электрического тока через материал детали, электромагнитным 

излучением или химическими реакциями. С учетом уравнения для плотности 

теплового потока, уравнение Пуассона можно записать в следующем виде: 

Задача 1. Равномернонагреваемый стержень полукруглого сечения 

В результате протекания электрического тока происходит нагрев 

длинного полукруглого стержня из нержавеющей стали. Этот стержень 

насположен внутри гранитного блока, который охлаждается смесью из воды 

и льда. Примем, что мощность объемного тепловыделения в стержне равна 

1e6, а в блоке – 0. Граничные условия в рассматриваемом случае 

соответствуют заданию постоянной температуры на внешней повенрхности 

гранитного блока. Программа для расчета распределения температуры в 

стержне и в блоке приведена ниже. 

 

TITLE 'Heated Semi-Circular Rod'  

SELECT errlim=1e-5 spectral_colors 

VARIABLES temp 

DEFINITIONS { SI units (m, kg, s, K, W) throughout } 

Lx=0.1 Ly=0.1 r0=0.05 

k heat { Declared only }fluxd_x=-k*dx(temp) fluxd_y=-k*dy(temp) 

fluxd=vector( fluxd_x, fluxd_y) fluxdm=magnitude( fluxd) 

EQUATIONS 

dx( fluxd_x)+ dy( fluxd_y)- heat=0 

BOUNDARIES 

region 'granite' k=3.5 heat=0 { Assign for entire domain } 

start 'outer' (-Lx,-Ly) value(temp)=273 { Ice-water } 

line to (Lx,-Ly) to (Lx,Ly) to (-Lx,Ly) to close 

region 'steel' k=45 heat=1e6 { Re-assign for steel } 

start 'rod' (-r0,0) line to (r0,0) arc to (0,r0) close 

PLOTS 

contour( temp) surface( temp) 

vector(fluxd) norm contour(fluxdm) painted 

elevation( normal(fluxd)) on 'rod' 

elevation( normal(fluxd)) on 'outer' 

report( pi*r0^2/2* 1e6) as 'Heating power' 

END 

 
Из графика распределения температуры (рис. 9.1) видно, что внутри 

стержня температура имеет максимальное значение. 

Рис. 9.1. 

 

Векторный график (рис. 9.2) показывает направление переноса тепла – 

из стержня, имеющего наибольшую температуру, в гранитный блок  и далее 

в окружающую этот блок среду, которая поддерживается при постоянной 

температуре. Из рис. 9.2 хорошо видно, что максимальная скорость 

теплопереноса наблюдается вблизи угловых точек. 

 

 
Рис. 9.2. 

 

Задача 2. Термоупругость 

В результате неравномерного нагрева твердого тела и связанного с 

этим теплового расширения могут возникать термонапряжения. В случае 

неравномерного распределения температуры закон Гука, для деформации в 

направлении координатной оси x, можно записать в следующеми виде 

Аналогичные соотношения могут быть записаны и для координатной 

оси y. Здесь, α – коэффициент линейного расширения; ΔT – отклонение 

температуры от заданного значенния, например, от комнатной температуры. 

В случае плоского напряженного состояния (σz = 0) эти уравнения 

можно упростить 

 
 

Разрешая эти уравнения относительно напряжений, получим: 

 

 
Касательные напряжения (напряжения сдвига) τxy в явном виде не 

зависят от температуры ΔT, и поэтому выражением для касательных 

напряжений (τ xy) будет таким же, как и для тела, имеющего одинаковую 

температуру. 

 

Задача 3. Стальной блок с постоянным градиентом температуры 

Рассмотрим теперь гипотетический пример: стальной блок расположен 

на абсолютно жестком основании, имеющем температуру 300 K, 

противоположная свободная поверхность поддерживается при температуре 

600 K. Левая и правая границы теплоизолированы, но могут свободно 

перемещаться. Для этого примера нужно использовать включаемый файл 

defuv.pde. 

 

{ Include file for thermo-elasticity in (x,y) } { defuvt.pde } 

C=E/(1-mu^2) G=E/[2*(1+mu)] 

uv=vector(u,v) uvm=magnitude( uv) 

ex=dx(u) ey=dy(v) exy=dx(v)+ dy(u) 

sx=C*(ex+ mu*ey- (1+mu)*alpha*dtemp) 

sy=C*(mu*ex+ ey- (1+mu)*alpha*dtemp) sxy=G*exy 

fluxd_x=-k*dx( dtemp) fluxd_y=-k*dy( dtemp) 

fluxd=vector( fluxd_x, fluxd_y) fluxdm=magnitude( fluxd) 

p_ang=0.5* arctan( 2*sxy/(sx-sy) ) { Radians } 

sxp0=(sx+sy)/2+ (sx-sy)/2*cos(2*p_ang)+ sxy*sin(2*p_ang) 

syp0=(sx+sy)/2- (sx-sy)/2*cos(2*p_ang)- sxy*sin(2*p_ang) 

{ Test for highest algebraic value: } 

p_angl= if sxp0>syp0 then p_ang else p_ang+ pi/2 

sxp= if sxp0>syp0 then sxp0 else syp0 

syp= if sxp0>syp0 then syp0 else sxp0 

p_angle=p_angl*180/pi { Degrees } 

mises=sqrt( 0.5*[ (sx-sy)^2+ sx^2+ sy^2]+ 3*sxy^2) 

energy_d=(1/2)*( sx*ex+ sy*ey+ sxy*exy) { Energy density } 

 

Так как решение задачи теплопереноса имеет элементарное решение, 

ниже дается решение только для расчета термоперемещений. 

 

TITLE 'Block Strained by a Temperature Gradient' { mex112.pde } 

SELECT errlim=1e-5 spectral_colors 

VARIABLES u v 

DEFINITIONS 

Lx=100e-3 Ly=200e-3 

mu=0.3 E =200e9 { Steel } 

alpha=1e-5 k=45 { Thermal properties } 

dtemp=y/Ly*300 { Temperature above ambient } 

#include 'defuvt.pde' 

EQUATIONS 

u: dx( sx)+ dy( sxy)=0 

v: dx( sxy)+ dy( sy)=0 

BOUNDARIES 

region 'steel' 

start 'outer' (-Lx,0) value(u)=0 value(v)=0 line to (Lx,0) 

load(u)=0 load(v)=0 line to (Lx,Ly) to (-Lx,Ly) close 

PLOTS 

grid(x+100*u, y+100*v) contour(dtemp) 

contour( u) contour( v) vector( uv) norm 

contour( sx) painted contour( sy) painted contour( sxy) painted 

elevation( u, v) on 'outer' 

elevation( sx) on 'outer' elevation( sy) on 'outer' 

contour( p_angle) 

vector( cos(p_angl)*sxp, sin(p_angl)*sxp) norm notips as ' "tension" ' 

contour( sxp) contour( syp) contour( mises) 

contour( sxp) painted contour( syp) painted 

contour( mises) painted 

END 

 



69 

Деформированная сетка показывает изменение поперечного сечения 

из-за теплового расширения верхней части блока. 

 

 
Рис. 9.3. 

 

На следующем рисунке представлено распределение напряжений вдоль 

оси x – sx. Из рисунка видно, что напряжения равны нулю на верхней, лево и 

правой границах блока, что соответствует заданным граничным условиям. 

Другие составляющие напряжений также равны нулю на соответствующих 

участках границы. 

 
Рис. 9.4. 

 

Значения sx на большей частьи блока являются отрицательными, что 

означает напряжения сжатия, однако в ограниченной области вблизи от 

основания действуют растягивающие напряжения. 

3. ОБЪЕКТ И СРЕДСТВА ИССЛЕДОВАНИЯ 
В качестве средства исследования служит прикладная программа для 

численного конечно-элементного анализа FlexPDE, объектом исследования 

является напряженное состояние пластины, через внутреннюю часть которой 

пропускается электрический ток. 
 
 

4. ПОРЯДОК ПРОВЕДЕНИЯ ИССЛЕДОВАНИЯ. 
1. Изучить теоретические сведения. 
2. Выполнить текстовые расчеты по программам указанным 

преподавателем. 
3. Ввести в систему FlexPDE текст программы, приведенный в 

теоретической части, и выполнить расчеты. Внести изменения в текст 
программы по задания преподавателя 

4. Подготовить отчет о выполнении работы. 
 
 
 

5. ФОРМА ОТЧЕТНОСТИ 
Отчет формируется в виде файла WinWord, в который включаются 

описания основных разделов описания задач в среде FlexPDE, текст 
прикладной программы и полученные результаты моделирования. 

 

6. КОНТРОЛЬНЫЕ ВОПРОСЫ 
1. Что понимается под плотностью теплового потока 
2. При каких условиях в твердом теле могут возникать 

термонапряжения? 
3. Что можно предпринять для уменьшения термонапряжений? 
4. Какие физические или химичсекие процессы могут быть связаны с 

объемным тепловыделением? 

